

Agent Based Modelling: Study of wealth repartition with a modified simple economy model

Benjamin Lepers, DSTI

June 5, 2019

Abstract

With a modified economy model from Netlogo[2] we investigate how the generosity of the agents affects the wealth distribution on the agent population. We have divided the agent population in 3 classes: poor, medium and rich. We study two distinct scenarios:
a) The rich class gives more than the medium classes which gives more than the poor class
b) The rich class gives less than the medium class which gives less than the poor class
As expected the scenario a) achieves less wealth inequalities than scenario b).

1 The original simple economy model

In the original economy model[3] with the defaults settings, 500 agents are given an initial wealth of 100 \$ each.

Each agent will give 1 \$ to an other agent randomly (if its wealth is positive). Each time an agent gives 1 \$, its wealth decreases by 1 \$. Hence the total amount of money, $500 \times 100 = 50000$ \$ is conserved (see the paper [1]).

The metrics of the wealth population are the total wealth of the top 10 % wealthier and the total wealth of the bottom 50 % poorer of the whole population.

We run the experiment 10 times using the behaviourspace option of Netlogo. After running 10000 ticks, an averaged wealth for the top 10 % and bottom 50 % population are respectively 12690 and 9990 \$ and the wealth evolution over time is given in figure 1.




Figure 1: Evolution of wealth over time of the top 10 % and bottom 50 % agents for the original economic model. Starting from a uniform distribution, and with the rule that each agent gives 1 \$ to another agent when possible, after some time, the wealth of the top 10 % is larger than the wealth of the bottom 50 %




Figure 2: distribution of wealth over time for the top 10 % and bottom 50 % group for the fair economic model. In this scenario, the wealth of the top 10 % group is not larger as the wealth of the bottom 50 % group

2 Modified model, rich give more than poor

The model is based on the economy model[3] from Netlogo. We add the rich, medium and poor classes with the following lines:

```
ask turtles with [ (wealth > 0 ) and (wealth <= 30) ] [transact_poor ]
ask turtles with [ (wealth > 30 ) and (wealth <= 80) ] [transact_medium]
ask turtles with [ wealth > 80 ] [transact_rich]
```

Agents with wealth less than 30 \$ are considered as poor, medium wealth is between 30 and 80 \$. The rich class is made of agent with wealth higher than 80 \$. The bounds are somewhat arbitrary. In this scenario, the rich gives 2 times more than the medium group and 10 times more than the poor group.

```
to transact_poor
  set wealth wealth - 1
  ask one-of other turtles [ set wealth wealth + 1 ]
end
to transact_medium
  set wealth wealth - 5
  ask one-of other turtles [ set wealth wealth + 5 ]
end
to transact_rich
  set wealth wealth - 10
  ask one-of other turtles [ set wealth wealth + 10 ]
end
```

In the report indicator, to check the money conservation we have also added the wealth of the rest of the agents, which have a wealth between the poorest bottom 50 % and the top 10 %.

```
to-report rest-wealth
  report sum [wealth] of max-n-of (count turtles) turtles [ wealth] -
    sum [ wealth ] of max-n-of (count turtles * 0.10) turtles [ wealth ]-
    sum [ wealth ] of min-n-of (count turtles * 0.50) turtles [ wealth ]
end
```

After running 10 times the simulation over 10 000 ticks with the behaviour space option, the average wealth for the top 10 % and bottom 50 % agent population are respectively 8652 and 19293 \$. The top 10 % wealth do not exceed the bottom 50 % wealth. The figure 2 shows that the system reaches quickly a steady state regime.




Figure 3: distribution of wealth over time for the top 10 % and bottom 50 % group for the unfair economic model. In this scenario after 20000 ticks, the wealth of the top 10 % group is 1.5 larger than all the rest of the agent population

3 Modified model, poor give more than rich

This model is almost the same as the previous one except that we change the generosity of the agents. Here the rich group gives 2 times less than the medium wealth group and 5 times less than the poor group as shown in the code below:

```
to transact_poor
  set wealth wealth - 5
  ask one-of other turtles [ set wealth wealth + 5 ]
end
to transact_medium
  set wealth wealth - 2
  ask one-of other turtles [ set wealth wealth + 2 ]
end
to transact_rich
  set wealth wealth - 1
  ask one-of other turtles [ set wealth wealth + 1 ]
end
```

The wealth evolution is shown in figure 3. After running 10 simulations over 10000 ticks with the behaviour space, the average wealth for the top 10 % and bottom 50 % are respectively 23884 and 727 \$. These numbers show a very strong wealth inequality between the agents.

4 Conclusion

The original economy model shows that when agents are giving 1 \$ randomly to other agents and with the additional rule that no negative wealth is possible (overall money conservation) a rich class emerges ({12.7, 10} k\$ respectively for the top 10 % and bottom 50 %).

With a modified model that includes a rich, medium and poor class, the wealth distribution is modified according to the generosity of each class (or one could also say according to the tax system implemented in the economy of a given state).

As expected, if the rich group gives more than the medium group and the medium group more than the poor group, a very rich class does not emerge and the wealth is more uniformly distributed after reaching a steady state regime ({8.6, 19.2} k\$ respectively for the top 10 % and bottom 50 %).

On the opposite, if we have a world with less generosity (the rich group is less generous than the medium group which is less than the poor group), then a rich class emerges

much quickly than the original uniform 1 \$ scenario ({23.8, 0.7} k\$ respectively for the top 10 % and bottom 50 %)

To improve the realism of these models, more complex rules could be implemented, such as allowing negative wealth (agents in debts) and introducing interests rate. In this case, the total amount of money is not more constant, and the analysis of wealth is more complex.

References

- [1] A. Dragulescu and V. Yakovenko. Statistical mechanics of money. *European Physics Journal B*, 2000.
- [2] U. Wilensky. Netlogo. <http://ccl.northwestern.edu/netlogo/>, 1999. Center for Connected Learning and Computer-Based Modeling, Northwestern University, Evanston, IL.
- [3] U. Wilensky. Netlogo simple economy models. <http://ccl.northwestern.edu/netlogo/models/SimpleEconomy>, 2011. Center for Connected Learning and Computer-Based Modeling, Evanston, IL.

Appendix A Fair economy model

```
turtles-own [ wealth ]

to setup
  clear-all
  create-turtles 500 [
    set wealth 100
    set shape "circle"
    set color green
    set size 2
    ;; visualize the turtles from left to right in ascending order of wealth
    setxy wealth random-ycor
  ]
  reset-ticks
end

to go
  ;; transact and then update your location
  ask turtles with [ (wealth > 0 ) and (wealth <= 30) ] [transact_poor ]
  ask turtles with [ (wealth > 30 ) and (wealth <= 80) ] [transact_medium]
  ask turtles with [ wealth > 80 ] [transact_rich]
  ;; prevent wealthy turtles from moving too far to the right
  ask turtles [ if wealth <= max-pxcor [ set xcor wealth ]]
  tick
end

to transact_poor
  set wealth wealth - 1
  ask one-of other turtles [ set wealth wealth + 1 ]
end

to transact_medium
```

```

    set wealth wealth - 5
    ask one-of other turtles [ set wealth wealth + 5 ]
end

to transact_rich
    set wealth wealth - 10
    ask one-of other turtles [ set wealth wealth + 10 ]
end

to-report top-10-pct-wealth
    report sum [ wealth ] of max-n-of (count turtles * 0.10) turtles [ wealth ]
end

to-report bottom-50-pct-wealth
    report sum [ wealth ] of min-n-of (count turtles * 0.50) turtles [ wealth ]
end

to-report rest-wealth
    report sum [wealth] of max-n-of (count turtles) turtles [ wealth] -
    sum [ wealth ] of max-n-of (count turtles * 0.10) turtles [ wealth ]-
    sum [ wealth ] of min-n-of (count turtles * 0.50) turtles [ wealth ]
end
; Copyright 2011 Uri Wilensky.
; See Info tab for full copyright and license.

```

Appendix B Unfair economy model

```

turtles-own [ wealth ]

to setup
    clear-all
    create-turtles 500 [
        set wealth 100
        set shape "circle"
        set color green
        set size 2

        ;; visualize the turtles from left to right in ascending order of wealth
        ;; in the model setting box, we activate the option world wraps horizontally
        ;; otherwise message error.
        setxy wealth random-ycor
    ]
    reset-ticks
end

to go
    ;; transact and then update your location
    ask turtles with [ (wealth > 5 ) and (wealth <= 30) ] [transact_poor ]
    ask turtles with [ (wealth > 30 ) and (wealth <= 80) ] [transact_medium]
    ask turtles with [ wealth > 80 ] [transact_rich]
    ;; prevent wealthy turtles from moving too far to the right

```

```

    ask turtles [ if wealth <= max-pxcor [ set xcor wealth ]]
    tick
end

to transact_poor
    set wealth wealth - 5
    ask one-of other turtles [ set wealth wealth + 5 ]
end

to transact_medium
    set wealth wealth - 2
    ask one-of other turtles [ set wealth wealth + 2 ]
end

to transact_rich
    set wealth wealth - 1
    ask one-of other turtles [ set wealth wealth + 1 ]
end

to-report top-10-pct-wealth
    report sum [ wealth ] of max-n-of (count turtles * 0.10) turtles [ wealth ]
end

to-report bottom-50-pct-wealth
    report sum [ wealth ] of min-n-of (count turtles * 0.50) turtles [ wealth ]
end

;; to check the money conservation
to-report rest-wealth
    report sum [wealth] of max-n-of (count turtles) turtles [ wealth ] -
    sum [ wealth ] of max-n-of (count turtles * 0.10) turtles [ wealth ] -
    sum [ wealth ] of min-n-of (count turtles * 0.50) turtles [ wealth ]
end

; Copyright 2011 Uri Wilensky.

```